

February 2007

# Variational Calculation of the Hydrogen Molecular Cation Using Maple

Carl W. David

*University of Connecticut*, [Carl.David@uconn.edu](mailto:Carl.David@uconn.edu)

Follow this and additional works at: [https://opencommons.uconn.edu/chem\\_educ](https://opencommons.uconn.edu/chem_educ)

---

## Recommended Citation

David, Carl W., "Variational Calculation of the Hydrogen Molecular Cation Using Maple" (2007). *Chemistry Education Materials*. 35.  
[https://opencommons.uconn.edu/chem\\_educ/35](https://opencommons.uconn.edu/chem_educ/35)

# Variational Calculation of the Hydrogen Molecular Cation ( $H_2^+$ ) using Maple

C. W. David

Department of Chemistry

University of Connecticut

Storrs, Connecticut 06269-3060

(Dated: February 28, 2007)

## I. SYNOPSIS

It is of interest to carry out a full calculation of the  $H_2^+$  quantum mechanical energy as a function of the internuclear distance so that one understands that the 2-electron problem's difficulties are, at least partially, due to the electron-electron repulsion.

## II. INTRODUCTION

Elementary texts show the ground state wave function of  $H_2^+$  to be

$$1\sigma \sim 1s_A + 1s_B$$

where  $1s_i$  means a  $1s$  orbital located on nucleus  $i$ .




FIG. 1: The Elliptical Coordinate System for Diatomic Molecules. The  $\mu$  coordinate is not depicted.

Next, the two wave functions in functional[1] rather than in “symbolic” coordinates:

$$\psi_A = e^{-kr_A} = 1s_A; \psi_B = e^{-kr_B} = 1s_B$$

where

$$r_A = \sqrt{x^2 + y^2 + (z - R/2)^2}$$

and

$$r_B = \sqrt{x^2 + y^2 + (z + R/2)^2}$$

( $R$  = internuclear distance, with nuclei at  $(0,0,\pm R/2)$ ). Finally, we re-write the wave function in elliptical coordinates: where

$$\lambda = \frac{r_A + r_B}{R}$$

and

$$\mu = \frac{r_A - r_B}{R}$$

giving (in Maple format NOTE, the following is WRONG. It is being left in for use in a class exercise in finding and eliminating errors in coding):

```

=====
assume(k>0,R>0);
psi_A := exp(-k*(R/2)*(lambda-mu));
psi_B := exp(-k*(R/2)*(lambda+mu));
=====

```

$$psi_A := e^{(-1/2 k R (\lambda - \mu))}$$

$$psi_B := e^{(-1/2 k R (\lambda + \mu))}$$

The tildes are a result of the “assume( $k > 0, R > 0$ )” statement, i.e., the assumption to be used that these two “variables” are to be regarded as positive definite. They are necessary so that future integrations will “work”.

## III. THE OVERLAP INTEGRALS

$$S_{AA} = \int_{all\ space} \psi_A^2 d\tau$$

where, in elliptical coordinates,

$$d\tau = \frac{R^3}{8} d\phi (\lambda^2 - \mu^2) d\lambda d\mu$$

The term (t1) corresponds to the S(AA) *integrand* in elliptical coordinates.

```
#####
t1 := (psi_A**2)*2*Pi*
((R**3)/8)*(lambda**2-mu**2):
#####
```

$2*\pi$  is the result of  $d\phi$  integration, and  $((R**3)/8)(\lambda**2-\mu**2) d\lambda d\mu$  is the rest of the volume element which, along with  $d\phi$  corresponds to  $dx dy dz$ .

#### A. Integration of Mono-Nuclear Overlap ( $S_{AA}$ and $S_{BB}$ )

```
#####
t2 := expand(int(t1,mu=+1..-1)):
#####
```

This line (creating t2) corresponds to integrating over  $\mu$  from +1 to -1 (as we know this is the domain), and then expanding the result,

```
#####
SAA := int(t2,lambda=1..infinity);
#####
```

we finally carry out the second ( $\lambda$ ) integral (arriving at the overlap integral  $S_{AA}$ ).

$$SAA := \frac{\pi}{k^{-3}}$$

Just as a check, we integrate over the B orbital, and obtain the same normalization constant.

```
#####
t11 := (psi_B**2)*2*Pi*((R**3)/8)*
(lambda**2-mu**2):
t21 := expand(int(t11,mu=+1..-1)):
t21 := collect(t21,exp(k*R*lambda)):
SBB := int(t21,lambda=1..infinity);
#####
```

and here is the output, showing that  $S_{AA} = S_{BB}$ :

$$SBB := \frac{\pi}{k^{-3}}$$

Notice that neither of these integrals is a function of  $R$ .

#### B. Setting Up the A-B Overlap Integral

Now, we attempt the overlap integral itself. Notice the product of wave functions.

$$\psi_{AB} = \int_{all\ space} \psi_A \psi_B d\tau$$

```
t3 := 2*Pi*psi_A*psi_B*((R**3)/8)*
(lambda**2-mu**2):
```

This is the famous  $S_{AB}$  integral.

```
#####
t4 := expand(int(t3,mu=+1..-1)):
t4 := collect(t4,exp(k*R*lambda)):
SAB := expand(int(t4,lambda=1..infinity)):
SAB := collect(SAB,exp(k*R));
#####
```

$$SAB := \frac{\frac{1}{3} \frac{\pi R^{-2}}{k^{-}} + \frac{\pi R^{-}}{k^{-2}} + \frac{\pi}{k^{-3}}}{e^{(k^{-} R^{-})}}$$

Here, we notice that the resultant overlap is a function of  $R$ .

#### IV. ENERGY OPERATORS

Here we introduce the Laplacian in elliptical coordinates. (See [http://digitalcommons.uconn.edu/chem\\_educ/5](http://digitalcommons.uconn.edu/chem_educ/5))

$$\nabla^2 = \frac{4}{R^2(\lambda^2 - \mu^2)} \left\{ \left( \frac{\partial((\lambda^2 - 1)\frac{\partial}{\partial\lambda})}{\partial\lambda} \right) + \left( \frac{\partial((1 - \mu^2)\frac{\partial}{\partial\mu})}{\partial\mu} \right) + \left( \frac{\partial\left(\frac{\lambda^2 - \mu^2}{(\lambda^2 - 1)(1 - \mu^2)}\frac{\partial}{\partial\phi}\right)}{\partial\phi} \right) \right\}$$

First, we form  $\nabla^2\psi_A$ :

```
#####
#start HAA section:
t5 := 4/((R**2)*(lambda**2-mu**2))*
(diff((lambda**2-1)*diff(psi_A,lambda),lambda)+
diff((1-mu**2)*diff(psi_A,mu),mu)):
#####
```

After operating with the Laplacian, we attempt the integration by, first, left multiplying with the A-orbital (later we will left multiply by the B-orbital), and then integrating, i.e.,

$$\int_{all\ space} \psi_A \nabla^2 \psi_A d\tau$$

```
#####
t6 := psi_A* 2*Pi*((-hbar**2)/(2*m))
*t5*(R**3)/8*(lambda**2-mu**2):
KEAA := int(t6,mu=+1..-1):
KEAA := int(KEAA,lambda=1..infinity):
KEAA := collect(KEAA,exp(k*R));
#####
```

$$KEAA := \frac{1}{2} \frac{\pi \hbar^2}{k^{-} m}$$

### A. Part of the Exchange Integral

This is now the part where we operate with the Laplacian on the B-orbital, but left multiply by the A-orbital.

$$KE_{AB} = \int_{all\ space} \psi_A \nabla^2 \psi_B d\tau$$

First, we evaluate

$$\nabla^2 \psi_B$$

```
#####
#HAB section
t5B := 4/((R**2)*(lambda**2-mu**2))*
(dif((lambda**2-1)*dif(psi_B,lambda),lambda)+
dif((1-mu**2)*dif(psi_B,mu),mu)):
#####
```

and then left multiplying by  $\psi_A$  we obtain

```
#####
t61 :=psi_A* 2*Pi*((-hbar**2)/(2*m))*
t5B*(R**3/8)*(lambda**2-mu**2):
#####
```

t61 is, along with atomic constants, t5B left multiplied by psi\_A.

```
#####
KEAB := int(t61,mu=+1..-1):
KEAB := collect(KEAB,exp(k*R)):
KEAB := int(KEAB,lambda=1..infinity):
KEAB := collect(KEAB,exp(k*R)):
#####
```

The final kinetic energy term  $KE_{AB}$ ,

$$KE_{AB} := -\frac{1}{6} e^{(-k^{\sim} R^{\sim})} \frac{\pi \hbar^2 (-3 + k^{\sim 2} R^{\sim 2} - 3 k^{\sim} R^{\sim})}{k^{\sim} m}$$

### B. The Potential Energy Operator

We need to integrate

$$\int_{all\ space} \psi_i * \frac{Z_A e^2}{r_A} * \psi_j d\tau$$

and

$$\int_{all\ space} \psi_i * \frac{Z_B e^2}{r_B} * \psi_j d\tau$$

for all i and j, dividing by  $r_A = (R/2) * (\lambda - \mu)$  will give one kind of cancellation, while dividing by  $r_B = (R/2) * (\lambda + \mu)$  will give the other kind of cancellation.

$(\lambda^2 - \mu^2) = (\lambda + \mu) * (\lambda - \mu)$ , so when dividing by  $r_A = (R/2) * (\lambda - \mu)$  one loses one term of this product i.e.,

$$\frac{\frac{R^3}{8}(\lambda^2 - \mu^2)}{\frac{R}{2}(\lambda \pm \mu)} = \frac{R^2}{4}(\lambda \mp \mu)$$

while when dividing by the other one loses the other.

```
#####
I1_a := -2*Pi*Z_A*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda+mu);
#####
```

I1\_a is the Coulomb attraction term (1/r\_B) concerning nucleus A

```
#####
I2_a := -2*Pi*Z_B*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda-mu);
#####
```

I2\_a is the same kind of term concerning nucleus B.

```
#####
I1_b := int(I1_a,mu=+1..-1):
I2_b := int(I2_a,mu=+1..-1):
I1_c := int(I1_b,lambda=1..infinity):
I2_c := int(I2_b,lambda=1..infinity):
H_AA := KEAA+I1_c+I2_c:
#####
```

```
#####
I3_a := -2*Pi*Z_A*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda+mu):
#####
```

I3\_a is the Coulomb attraction term (1/r\_B) concerning nucleus A

```
#####
I4_a :=
-2*Pi*Z_B*esq*((R**2)/4)*
exp(-k*R*lambda)*(lambda-mu):
#####
```

I4\_a is the Coulomb attraction term (1/r\_A) concerning nucleus B.

```
#####
I3_b := int(I3_a,mu=+1..-1):
I4_b := int(I4_a,mu=+1..-1):
I3_c := int(I3_b,lambda=1..infinity):
I4_c := int(I4_b,lambda=1..infinity):
#####
```

We now form the two states, bonding and anti bonding:

```
#####
H_AB := KEAB + I3_c+I4_c:
Energy_one := (H_AB-H_AA)/(SAB-SAA):
Energy_two := (H_AB + H_AA)/(SAB+SAA):
#####
```

Here we specialize to  $H_2^+$ :

```
#####
Energy_one := subs(Z_A=1,Z_B=1,Energy_one):
Energy_two := subs(Z_A=1,Z_B=1,Energy_two):
#####
```

Here we transform to atomic units:

```
#####
Energy_one :=
subs(esq=1,hbar=1,m=1,Energy_one):
Energy_two := subs(esq=1,hbar=1,m=1,Energy_two):
#####
```

(Note, please, that the following lines have been edited to fit on the page properly.)

$$Energy\_two(numerator) := -\frac{1}{6k}e^{(-kR)}\pi \left( (-3 + k^2R^2 - 3kR) - 4\frac{kR + 1}{k} \right) + \frac{\pi}{2k}$$

$$Denominator = \frac{1}{3} \left( \frac{\pi R^2}{k} + \frac{\pi R}{k^2} + \frac{\pi}{k^3} \right) e^{(-kR)} + \frac{\pi}{k^3}$$

```
#####
plot3d(\{Energy_one+1/R,Energy_two+1/R\}
,k=1.5..2.5
,R=0.2..4,axes=BOXED);
#####
```

```
#####
En := subs(R=0.5,Energy_two+1/0.5):
plot(En,k=0.0..2.0,labels=['k','Energy'],
title='Variation of Exponential k');
#####
```

## V. DISCUSSION OF FIGURES

We suggest (above) how to do the plotting of the results of these calculations, but, below, display plots which

```
> restart;
> with(plots):
> assume(k>0,R>0);
> psi_A := exp(-k*(R/2)*(lambda-mu));
> psi_B := exp(-k*(R/2)*(lambda+mu));
> r_A := (R/2)*(lambda-mu);
> r_B := (R/2)*(lambda+mu);
> dtau := ((R^3)/8)*(lambda^2-mu^2);

> #####
> t1 := (psi_A**2)*2*Pi*((R**3)/8)*(lambda**2-mu**2):
> t2 := expand(int(t1,mu=-1..1)):
> t2 := collect(t2,exp(k*R*lambda)):
> print('here is SAA');
> SAA := int(t2,lambda=1..infinity);
> psi_A := psi_A/sqrt(SAA);#normalize
```

are not coded. For pedagogical purposes, one can only stress that involvement with the code itself is the only way to appreciate what is going on, and just reading this or that commentary is only partially enlightening. Mastery comes with effort.

The plots shown home in on two facts about these computations. First, the electronic energy of the  $H_2^+$  molecular cation is a function of the internuclear distance, and splitting diagrams which show 1s orbitals splitting into  $\sigma$  and  $\sigma^*$  are diagrams at a fixed R value, generally where the energy of the bonding state is a minimum! But Figure 3 we can see clearly the splitting between the  $1\sigma$  and the  $1\sigma^*$  in the vicinity of  $R \approx 2au$ .

Second, it is interesting to note how weak the minimum is in the LCAO-MO approximation. In fact, Maple itself could not find the minimum! Hence the approximate value employed (0.78) for k.

Finally, we note that the entire exercise is wrong. The limits as  $R \rightarrow \infty$  are positive, while they should be negative, since at that limit, we have an H atom and a proton, and the H atom's electron is in the 1s state! Figure 3 clearly shows that at  $R > 5$  the energies are asymptotically positive!

Further, 4 gives no hint that there is an error here, i.e., the seduction of good looking graphs lowers our skepticism concerning accuracy!

After more trouble than I care to discuss, the following worked:

## VI. HERE IS THE CORRECTED VERSION

```

> t11 := (psi_B**2)*2*Pi*((R**3)/8)*(lambda**2-mu**2):
> t21 := expand(int(t11,mu=-1..1)):
> t21 := collect(t21,exp(k*R*lambda)):
> print ('here is SBB');
> SBB := int(t21,lambda=1..infinity);
> psi_B := psi_B/sqrt(SBB);
> t3 := 2*Pi*psi_A*psi_B*((R**3)/8)*(lambda**2-mu**2):
> t4 := expand(int(t3,mu=-1..1)):
> t4 := collect(t4,exp(k*R*lambda)):
> SAB := expand(int(t4,lambda=1..infinity)):
> SAB := collect(SAB,exp(k*R));
> #start HAA section:
> t5 := 4/((R**2)*(lambda**2-mu**2))*
> (diff((lambda**2-1)*diff(psi_A,lambda),lambda)+
> diff((1-mu**2)*diff(psi_A,mu),mu)):
> t6 :=psi_A* 2*Pi*((-hbar**2)/(2*m))*t5*dtau
> -2*Pi*Z_A*esq*(1/r_A)*psi_A*psi_A*dtau
> -2*Pi*Z_A*esq*(1/r_B)*psi_A*psi_A*dtau:
> HAA := int(t6,mu=-1..1):
> HAA := int(HAA,lambda=1..infinity);
> #HAB section
> t5B := 4/((R**2)*(lambda**2-mu**2))*
> (diff((lambda**2-1)*diff(psi_B,lambda),lambda)+
> diff((1-mu**2)*diff(psi_B,mu),mu)):
> t61 :=psi_A* 2*Pi*((-hbar**2)/(2*m))*t5B*dtau
> -2*Pi*Z_A*esq*(1/r_A)*psi_B*psi_A*dtau
> -2*Pi*Z_A*esq*(1/r_B)*psi_B*psi_A*dtau:
> HAB := int(t61,mu=-1..1):
> HAB := collect(HAB,exp(k*R)):
> HAB := int(HAB,lambda=1..infinity):
> HAB := collect(HAB,exp(k*R));
> Energy_one := (HAA-HAB)/(1-SAB):
> Energy_two := (HAB + HAA)/(SAB+1):
> #specialize to homonuclear Z=1 case
> Energy_one := subs(Z_A=1,Z_B=1,Energy_one);
> Energy_two := subs(Z_A=1,Z_B=1,Energy_two);
> Energy_one := subs(esq=1,hbar=1,m=1,Energy_one);
> Energy_two := subs(esq=1,hbar=1,m=1,Energy_two);

```

$$\text{Energy\_one} := \frac{1}{6} \frac{e^{(-k^- R^-)} \pi (-3 + k^{-2} R^{-2} - 3 k^- R^-)}{k^-} - \frac{1}{2} \frac{\pi}{k^-}$$

$$\frac{1}{3} \frac{\pi R^{-2}}{k^-} + \frac{\pi R^-}{k^{-2}} + \frac{\pi}{k^{-3}} - \frac{\pi}{k^{-3}}$$

```

> plot3d({Energy_one+1/R,Energy_two+1/R
> },k=1.0..1.5,R=0.2..4,axes=BOXED);

```

```

> En5 := subs(R=1.0,Energy_two+1/R):
> En6 := subs(R=1.2,Energy_two+1/R):
> En7 := subs(R=1.4,Energy_two+1/R):
> En8 := subs(R=1.6,Energy_two+1/R):
> En9 := subs(R=1.8,Energy_two+1/R):
> En10 := subs(R=2.0,Energy_two+1/R):
> En11 := subs(R=2.2,Energy_two+1/R):
> En12 := subs(R=2.4,Energy_two+1/R):
> En13 := subs(R=2.6,Energy_two+1/R):
> En14 := subs(R=2.8,Energy_two+1/R):
> En15 := subs(R=3.0,Energy_two+1/R):
> plot({En5,En6,En7,En8,En9,En10,En11,En12,En13,En14,En15
> },k=0.0..2.0,labels=['k','Energy'],title='Variation of Exponential
> k');
> contourplot(Energy_two+1/R,R=0.8..3.5,k=0.8..2.0,contours = 80);
> #minimize(Energy_two+1/R,R=0.8..3.5,k=0.8..2.0,location);#FAILED ON
> TIME
> #choose 1.2 (by eye);
> plot({subs(k=1.2,Energy_two+1/R),
> subs(k=0.78,Energy_one+1/R)
> },R=0.1..5.0,labels=['R','Energy'],title='E(R)');
> limit1 := limit(subs(k=1.2,Energy_one+1/R),R=infinity);
> limit2 := limit(subs(k=1.2,Energy_two+1/R),R=infinity);

```

Warning, the name changecoords has been redefined

$$psi\_A := e^{(-\frac{k^- R^- (\lambda - \mu)}{2})}$$

$$psi\_B := e^{(-\frac{k^- R^- (\lambda + \mu)}{2})}$$

$$r\_A := \frac{R^- (\lambda - \mu)}{2}$$

$$r\_B := \frac{R^- (\lambda + \mu)}{2}$$

$$dtau := \frac{R^{-3} (\lambda^2 - \mu^2)}{8}$$

here is SAA

$$SAA := \frac{\pi}{k^{-3}}$$

$$psi\_A := \frac{e^{(-\frac{k^- R^- (\lambda - \mu)}{2})} k^-}{\sqrt{\frac{\pi}{k^-}}}$$

here is SBB

$$SBB := \frac{\pi}{k^{-3}}$$

$$psi\_B := \frac{e^{(-\frac{k^- R^- (\lambda + \mu)}{2})} k^-}{\sqrt{\frac{\pi}{k^-}}}$$

$$SAB := \frac{\frac{1}{3} k^{-2} R^{-2} + k^- R^- + 1}{e^{(k^- R^-)}}$$

$$HAA := -\frac{1}{2} (2 Z\_A esq m (e^{(k^- R^-)})^2 R^- k^- - 2 Z\_A esq m - k^{-2} hbar^2 R^- (e^{(k^- R^-)})^2 - 2 Z\_A esq m k^- R^- + 2 (e^{(k^- R^-)})^2 Z\_A esq m) / (R^- (e^{(k^- R^-)})^2 m)$$

$$HAB := -\frac{1}{6} k^- (hbar^2 k^{-3} R^{-2} - 3 hbar^2 k^- + 12 Z\_A esq m - 3 hbar^2 k^{-2} R^- + 12 Z\_A esq m k^- R^-) e^{(-k^- R^-)} / m$$

$$\begin{aligned}
Energy\_two := & \left( -\frac{1}{6} \frac{k^\sim (\hbar bar^2 k^{\sim 3} R^{\sim 2} - 3 \hbar bar^2 k^\sim + 12 esq m - 3 \hbar bar^2 k^{\sim 2} R^\sim + 12 esq m k^\sim R^\sim) e^{(-k^\sim R^\sim)}}{m} \right. \\
& - \frac{1}{2} (2 esq m (e^{(k^\sim R^\sim)})^2 R^\sim k^\sim - 2 esq m - k^{\sim 2} \hbar bar^2 R^\sim (e^{(k^\sim R^\sim)})^2 - 2 esq m k^\sim R^\sim \\
& \left. + 2 (e^{(k^\sim R^\sim)})^2 esq m) / (R^\sim (e^{(k^\sim R^\sim)})^2 m) \right) / \left( \frac{\frac{1}{3} k^{\sim 2} R^{\sim 2} + k^\sim R^\sim + 1}{e^{(k^\sim R^\sim)}} + 1 \right)
\end{aligned}$$

$$\begin{aligned}
Energy\_one := & \left( -\frac{1}{2} \frac{2 (e^{(k^\sim R^\sim)})^2 R^\sim k^\sim - 2 - k^{\sim 2} R^\sim (e^{(k^\sim R^\sim)})^2 - 2 k^\sim R^\sim + 2 (e^{(k^\sim R^\sim)})^2}{R^\sim (e^{(k^\sim R^\sim)})^2} \right. \\
& + \frac{1}{6} k^\sim (k^{\sim 3} R^{\sim 2} - 3 k^\sim + 12 - 3 k^{\sim 2} R^\sim + 12 k^\sim R^\sim) e^{(-k^\sim R^\sim)} \left. / \left( \right. \right. \\
& \left. \left. 1 - \frac{\frac{1}{3} k^{\sim 2} R^{\sim 2} + k^\sim R^\sim + 1}{e^{(k^\sim R^\sim)}} \right) \right)
\end{aligned}$$

$$\begin{aligned}
Energy\_two := & \left( -\frac{1}{6} k^\sim (k^{\sim 3} R^{\sim 2} - 3 k^\sim + 12 - 3 k^{\sim 2} R^\sim + 12 k^\sim R^\sim) e^{(-k^\sim R^\sim)} \right. \\
& - \frac{1}{2} \frac{2 (e^{(k^\sim R^\sim)})^2 R^\sim k^\sim - 2 - k^{\sim 2} R^\sim (e^{(k^\sim R^\sim)})^2 - 2 k^\sim R^\sim + 2 (e^{(k^\sim R^\sim)})^2}{R^\sim (e^{(k^\sim R^\sim)})^2} \left. / \left( \right. \right. \\
& \left. \left. \frac{\frac{1}{3} k^{\sim 2} R^{\sim 2} + k^\sim R^\sim + 1}{e^{(k^\sim R^\sim)}} + 1 \right) \right)
\end{aligned}$$



## VII. WRONG FIGURES




FIG. 2: WRONG Energies of the ground and first excited states of the electron of the  $H_2^+$  cation

$limit1 := -0.4800000000$   
 $limit2 := -0.4800000000$

---

[1] C. W. David, *When are LCAO-MOs not Eigenfunctions?*,  
 J. Chem. Ed., 59, 288(1982).




FIG. 3: WRONG Energies of the ground and first excited states of the electron of the  $H_2^+$  cation assuming a fixed value of  $k=0.78$ , plotted as a function of the internuclear distance  $R$




FIG. 4: WRONG Contours of the Energy as a function of  $R$  and  $k$




FIG. 5: Energy as a function of  $R$  and  $k$  ( $\sigma$  and  $\sigma^*$ )




FIG. 6: Energy as a function of  $k$  ( $\sigma$ )




FIG. 7: Energy contour map as a function of  $R$  and  $k$  ( $\sigma$ )




FIG. 8: Energy as a function of  $R$  and  $k$  ( $\sigma$  and  $\sigma^*$ )